Croustillant
sur lit de calcul formel

TRIANGLE ÉQUILATÉRAL ET CUBIQUE



ABC est un triangle équilatéral.

idée : sur le site de Mathématica

à faire

- À l'aide d'un logiciel de géométrie dynamique, construire un triangle équilatéral ABC et son cercle inscrit, puis la cubique ayant pour racines les abscisses des sommets.
  Construire les perpendiculaire à l'axe des abscisses, tangentes au cercle inscrit.
  Émettre une conjecture.
- Démontrer cette conjecture.
- Réciproquement, à partir d'une cubique, construire un triangle équilatéral dont les abscisses des sommets sont les racines de la fonction.

Croustillant
sur lit de calcul formel

TRIANGLE ÉQUILATÉRAL ET CUBIQUE

On se place dans un repère tel que $A(0\,;y_A)$, $C(1\,;y_C)$ et $B(x_B\,;y_B)$

Avec Xcas : on construit la figure, puis les tangentes au cercle inscrit d'équations $x=\alpha$ et $x=\beta$. On vérifie que α et β sont les valeurs qui annulent la dérivée.

```
/* les abscisses des sommets du triangle équilatéral sont les racines de la cubique */
/* les tangentes au cercle inscrit ont pour équation x=a avec f'(a)=0... */
assume(ya=2)
A:=point(0,ya)
assume(yc=7)
C:=point(1,yc)
Tabc:=equilateral_triangle(A,C)
B:=vertices(Tabc)[2]
G:=isobarycenter(A,B,C)
[xg,yg]:=coordinates(G)
Cabc:=incircle(A,B,C)
[Ga,Gb]:=inter(line(y=yg),Cabc)
Ta:=line(x=abscissa(Ga))
Tb:=line(x=abscissa(Gb))

xb:=abscissa(B)
f(x):=x*(x-1)*(x-xb)
solve(f'(x)=0,x)
abscissa(Ga)
abscissa(Gb)
normal(f'(abscissa(Ga)))

```

On appelle G le centre de gravité du triangle et x_G l'abscisse de G.

Après analyse de la figure, démontrer que les droites d'équations $x=\alpha$ et $x=\beta$ sont tangente au cercle inscrit de ABC revient à démontrer que $x_G = \dfrac{\alpha+\beta}{2}$

On sait que G est le centre de gravité de ABC,
donc $\overrightarrow{GA} + \overrightarrow{GB} + \overrightarrow{GC} = \overrightarrow{0}$.

En passant aux abscisses : $-x_G + (x_B - x_G) + (1 - x_G) = 0$ d'où $x_G = \dfrac{x_B+1}{3}$

TRIANGLE ÉQUILATÉRAL ET CUBIQUE

On sait que $f(x) = x(x - x_B)(x - 1) = x^3 - (x_B + 1)x^2 + x_B x$
donc $f'(x) = 3x^2 - 2(x_B + 1)x + x_B$
on a donc $\alpha + \beta = \dfrac{2(x_B + 1)}{3}$ d'où $\dfrac{\alpha + \beta}{2} = \dfrac{x_B + 1}{3}$

Donc $x_G = \dfrac{\alpha + \beta}{2}$

pour continuer

Construire le triangle équilatéral ABC à partir de la courbe ?

On choisit un repère tel que 0 et 1 soit deux des trois racines. On appelle b l'autre et α et β les deux valeurs qui annulent la dérivée.

Les trois sommets ont pour coordonnées : $A(0\,;y_A)$, $B(x_B\,;y_B)$ et $C(1\,;y_C)$

G centre du triangle a pour coordonnées : $\begin{cases} x_G = \dfrac{1}{3}(x_B + 1) \\ y_G = \dfrac{1}{3}(y_A + y_B + y_C) \end{cases}$

donc G appartient à la droite d'équation $x = \dfrac{1}{3}(x_B + 1)$

A′ milieu de [BC] a pour coordonnées : $\begin{cases} x_{A'} = \dfrac{1}{2}(x_B + 1) \\ y_{A'} = \dfrac{1}{2}(y_B + y_C) \end{cases}$

donc A′ appartient à la droite d'équation $x = \dfrac{1}{2}(x_B + 1)$

la droite (GA′) est la médiane issue de A, donc le point A′ permet d'obtenir les points A, B et C.

$G'(\alpha\,;y_G)$ est un point du cercle inscrit, l'intersection de ce cercle avec la droite d'équation $x = \dfrac{1}{2}(x_B + 1)$ permet de trouver le point A′...

Commande

abscisse, assume, coordonnees, droite, fenêtre de géométrie, inscrit, inter, isobarycentre, normal, point, prime (′) solve, sommets, triangle_equilateral